
SIR VENDOR Demystified

With VSE/ESA, an assortment of new
operator commands were provided,

one of these being the “SIR” command.
Now “SIR” has a number of parameters,
including the following:

● SIR ? — displays the “Help” screen
● SIR SYS — displays VSE-related 

IPL information
● SIR RESET — the same as SIR SYS,

but resets counters/shows statistics
● SIR SMF(=<ON|OFF|cuu>) — 

the Subsystem Measurement Facility
● SIR MON(=<<id|ON(,NOSYM)> 

|OFF>(option)) — monitoring data
● SIR MIH(=<nnnnnn|ON|OFF>) — 

missing Interrupt Handler control
● SIR CHPID(=chpid) — displays 

CHPID information
● SIR CRWMSG(<ON|OFF>) — 

displays/modifies CRW message reporting
● SIR VENDOR — displays vendor 

product information

This month, I’ll examine the last command,
SIR VENDOR. Every once in awhile,
someone will learn about the SIR VENDOR
command and try it only to be disappointed.
Sure, this command provides you with a
description of the name of the vendor, the
name of the software product and some
release information, but only if the product
is active and only if the vendor provides
this information to the VSE system! For
example, there are certain VSE internal
functions that require the use of a special
PRODID (Product ID) macro that will let
you see whatever information the vendor
provides. However, not all products need
to use these special facilities, and there-
fore not all products will be displayed
when you enter SIR VENDOR on the
VSE console.

TESTPROG START 0
BALR 12,0
USING *,12
LA  4,PRODAREA
USING PRODSECT,4
MVC IJBCOMPN(14),=CL14’ABC Company’
MVC IJBPRODN(16),=CL16’LIBPACK’
MVC IJBVRM(6),=CL6’1.0a’
MVC IJBVIDL(2),=H’50’ 
PRODID DEFINE,AREA=(4)
PRODID DELETE,TOKEN=IJBTOKEN
EOJ  ,

PRODAREA DS   13F
PRODSECT  PRODID DSECT

END  

FIGURE 1: SAMPLE PROGRAM TO REGISTER PROGRAM WITH VSE

* * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * *           
* Program: QPRODID                                      *           
*                                                           *           
* Purpose: To list all of the products that have been defined       *           
*       To VSE/ESA using the standard interface. This version of    *           
*       QPRODID is limited to 256 entries.                          *           
*                                                               *           
* OUTPUT: A listing of those items is output to SYSLOG.             *           
* * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * *           

PUNCH ‘ PHASE QPRODID,*’                                               
QPRODID  START 0                       Start of program                         

BALR  R12,0                   Setup the base                           
USING *,R12                   Establish addressability                 
GETVIS ADDRESS=(R6),LENGTH=SIZE                                        
LTR  R15,R15                  Did it work ?                            
BZ   QD0010                   Yes...proceed                            
L    R3,=A(ERR4)              No...point to problem type               
BAL  R11,QD0080               Display the message                      
B    QD0070                   And exit                                 

QD0010   LR   R4,R6                    Point to work area                       
USING PRODSECT,R4             Map to layout                            

*                                                                               
* Let’s setup the fields required and get the data                              
*                                                                               

XC   IJBCOMPN(14),IJBCOMPN    Initialize the fields                    
XC   IJBPRODN(16),IJBPRODN    Prior to getting the data                
XC   IJBVRM(6),IJBVRM         To clear this information                

*                                                                               
MVC  IJBCKLEN(2),=H’10240’    Allow up to 256 entries                  
MVC  IJBVIDL(2),=H’50’        Setup the input area length              
PRODID CHECK,AREA=(R4)        Get the next pointer

*                                                                               
LTR  R15,R15                  Did it work ?                            
BZ   QD0040                   Yes...proceed                            
CH   R15,=H’4’                More than 256 entries ?                  
BNZ  QD0020                   No...it’s ok                             
LR   R15,R0                   Get the number of entries                
L    R3,=A(ERR2)              Point to the problem type                
B    QD0030                   And display the problem                  

QD0020   C    R15,=F’8’                Any records found ?                      
BZ   QD0045                   No...it’s an error                       
L    R3,=A(ERR1)              Point to the message info                

QD0030   BAL  R11,QD0080               Display the message                      
B    QD0070                   And terminate                            

*                                                                               

FIGURE 2: QPRODID 

For example, let’s say that I want to get
the ALET, a special access register token so
that I can access storage across partitions
without the need of an XMOVE macro. To
do so, I need to register my program with
VSE. Figure 1 presents an example of such
a program.  Actually, this example doesn’t
do a whole lot, but it does show you what a
program needs to do to define itself to the
system. The first thing you do is set up the
name of your company. Next, tell the system
the name of your product. I also define the
release information. Before you define this
information to the VSE system, set up the
length of the work field (50 bytes) and pass
control to the PRODID macro. After the call
works, it will save a special “token” in the
work area. You will want to use this special
token later to delete this information.

Of course, you should check the return
code (R15) after you perform such a program
call. One of the reasons that it may fail is if
a DEFINE already was performed elsewhere
with the same company name, product name
and release information. Every DEFINE
must be unique. It is for this reason that
software companies often limit the use of
this macro to programs that run in the SVA,
where there will be no conflict.

For a test, you may want to insert a
WTOR between the DEFINE and the

* Here we list all of the entries found                                         
*                                                                               
QD0040   LA   R7,50(R4)                Point to the output buffer               

CLI  0(R7),X’00’              Any entries ?                            
BNZ  QD0050                   Yes...proceed                            

QD0045   MVC  IOAREA(46),ERRM          Get the error message                    
EXCP CONSOLE                  Output to SYSLOG                         
WAIT CONSOLE                  Let the I/O complete                     
MVI  RESULT+3,X’04’           Indicate a minor problem                 
B    QD0070                   And exit                                 

QD0050   DS   0H                                                                
MVC  IOAREA(46),HEADER        Get the header                           
EXCP CONSOLE                  Output to SYSLOG                         
WAIT CONSOLE                  Let the I/O complete                     
MVC  IOAREA(46),DASHES        Get the delimiter                        
EXCP CONSOLE                  Output to SYSLOG                         
WAIT CONSOLE                  Let the I/O complete                     

QD0060   CLI  0(R7),X’00’              All done ?                               
BZ   QD0070                   Yes...exit                               
MVC  FIELD1(14),0(R7)         Get the company name                     
MVC  FIELD2(16),14(R7)        Get the product name                     
MVC  FIELD3(6),30(R7)         Get the version/release info             
MVC  IOAREA(46),PLINE         Get the detail line                      
EXCP CONSOLE                  Output to SYSLOG                         
WAIT CONSOLE                  Let the I/O complete                     
LA   R7,40(R7)                Skip a record                            
B    QD0060                   And do another one                       

* TEMINATE                                                                      
QD0070   L    R3,RESULT                Get the final return code                

EOJ  RC=(3)                   Return to VSE                            
*              
* This routine takes care of error message processing                           
*                                                                               
QD0080   CVD  R15,DBLWORK              Convert to decimal                       

UNPK RCODE(5),DBLWORK+5(3)    Get the amount                           
OI   RCODE+4,X’F0’            Get rid of any sign                      
MVC  ERRTYPE(15),0(R3)        Update the error message                 
MVC  IOAREA(46),ERR           Get the error line                       
EXCP CONSOLE                  Output to SYSLOG                         
WAIT CONSOLE                  Let the I/O complete                     
MVI  RESULT+3,X’08’           Setup for the return code                
BR   R11                      Return to caller                         

*  — — — — — — — — — — — — — — — — — — — —  DATA AREA — — — — — — — — — — — — — — — — — — —  *           
LTORG ,                       Literal pool goes here                   

IOAREA   DC   CL46’ ‘                                                           
CONSOLE  CCB  SYSLOG,CONCCW                                                     
CONCCW   CCW  1,IOAREA,0,46                                                     
*                                                                               
SIZE     DC   F’11264’                 Size used for GETVIS                     
RESULT   DC   F’0’                     Contains the return code                 
DBLWORK  DS   D                        For converting decimal                   
*                                                                               
* Message lines                                                                 
*                                                                               
HEADER   DC   CL14’Company Name’,CL3’ ‘,CL16’Name of Product’                   

DC   CL3’ ‘,CL6’VerMod’,CL8’ ‘                                         
*                                                                               
PLINE    DS   0CL46                                                             
FIELD1   DC   CL17’ ‘                                                           
FIELD2   DC   CL19’ ‘                                                           
FIELD3   DC   CL16’ ‘                                                           
*                                                                               
DASHES   DC   46CL1’-’                                                          
*                                                                               
* ERROR MESSAGES                                                                
*                                                                               
ERRM     DC   CL46’QPRODID could not find any installed products’               
*                                                                               
ERR      DC   CL21’QPRODID Error during ‘                                       
ERRTYPE  DC   CL17’ ‘,C’RC=’                                                    
RCODE    DC   CL11’ ‘                                                           
*                                                                               
ERR1     DC   CL15’PRODID check’                                                
ERR2     DC   CL15’O/P too small’                                               
ERR4     DC   CL15’GETVIS call’                                                 
* Equates                                                                       
R0       EQU  0                                                                 
R3       EQU  3                                                                 
R4       EQU  4                                                                 
R6       EQU  6                                                                 
R7       EQU  7                                                                 
R11      EQU  11                                                                
R12      EQU  12                                                                
R15      EQU  15                                                                
*  — — — — — — — — — — — — — — — — — —  DUMMY SECTION — — — — — — — — — — — — — — — —  *              
PRODSECT PRODID DSECT                  PRODID BUFFER                            

END    

FIGURE 2: CONTINUED

For example, there are
certain VSE internal

functions that require
the use of a special

PRODID (Product ID)
macro that will let you
see whatever information

the vendor provides.
However, not all products
need to use these special
facilities, and therefore

not all products will 
be displayed when you

enter SIR VENDOR 
on the VSE console.

©2000 Technical Enterprises, Inc. Reproduction of this document without permission is prohibited.



WWW.NASPA.NETTECHNICAL SUPPORT • FEBRUARY 2000

DELETE of the PRODID. In this way, you
can do a SIR VENDOR command and see
how much information is provided. 

Now, believe it or not, all of this is simply
text information stored in System GETVIS
and the token is a pointer. There is nothing
fancy about it. You can also write a small
program to locate and process the entire
chain of entries and display it in an easier
format. It is for this purpose that I wrote
QPRODID, as shown in Figure 2. You
might want to use a modified version of this
program to:

● prevent reinitializing a vendor product 
twice (already exists)

● prevent loading a program if 
a prequisite program does not exist

● automatically load vendor products that 
are not on the list

As you can see, we get the pointer to the
first entry in the chain and display it. We
then skip to the next one and the next one
until we run out of text entries. Granted,
you could make your own IJBTOKEN
pointer and issue a DELETE. However,

this could have a very bad effect on your
installed vendor products! 

